


Device Modeling Report

COMPONENTS : OPERATIONAL AMPLIFIER (CMOS)
PART NUMBER : NJU7044
MANUFACTURER : NEW JAPAN RADIO



Bee Technologies Inc.

Spice Model



```

*$
*PART NUMBER: NJU7044
*MANUFACTURER: NEW JAPAN RADIO
*CMOS OPAMP WITH SHUTDOWN
*All Rights Reserved Copyright (c) Bee Technologies Inc. 2006
.SUBCKT NJU7044 OUT1 -IN1 +IN1 VDD +IN2 -IN2 OUT2
+ OUT3 -IN3 +IN3 VSS +IN4 -IN4 OUT4
X_U1 +IN1 -IN1 VDD VSS OUT1 NJU7044_s
X_U2 +IN2 -IN2 VDD VSS OUT2 NJU7044_s
X_U3 +IN3 -IN3 VDD VSS OUT3 NJU7044_s
X_U4 +IN4 -IN4 VDD VSS OUT4 NJU7044_s
.ENDS NJU7044
.SUBCKT NJU7044_s IN+ IN- V+ V- OUT
m1 3 IN- 6 V- nix l=6u w=20.813u
m2 4 7 6 V- nix l=6u w=25u
m3 8 IN- 5 5 pix l=6u w=25u
m4 9 7 5 5 pix l=6u w=27.75u
eos 7 IN+ poly(1) 25 98 5e-3 0.451
iin1 IN+ 98 1.5p
iin2 IN- 98 1.5p
ios IN- IN+ 0.5p
i1 V+ 5 50u
i2 6 V- 50u
r1 V+ 3 4.833k
r2 V+ 4 4.833k
r3 8 V- 4.833k
r4 9 V- 4.833k
d3 5 V+ dx
d4 V- 6 dx
eref 98 0 poly(2) V+ 0 V- 0 0 0.5 0.5
g1 98 21 poly(2) 4 3 9 8 0 145u 145u
rg 21 98 53.2e6
cc 21 OUT 9.4p

```

```

d1 21 22 dx
d2 23 21 dx
v1 V+ 22 1.37
v2 23 V- 1.37
ecm 24 98 poly(2) IN+ 98 IN- 98 0 0.5 0.5
r5 24 25 1e6
r6 25 98 2.25k
c1 24 25 0.75p
isy V+ V- 390u
gsy V+ V- poly(1) V+ V- -3.334e-4 6.667e-5
ep V+ 39 poly(1) 98 21 0.78925 1
en 38 V- poly(1) 21 98 0.78925 1
m15 OUT 39 V+ V+ pox l=1.5u w=270u
m16 OUT 38 V- V- nox l=1.5u w=271u
c15 OUT 39 1p
c16 OUT 38 1p
c17 5 9 10p
c18 out 9 30p
.model dx d(rs=1 cjo=0.1p)
.model nix nmos(vto=.75 kp=205.5u rd=1 rs=1 rg=1 rb=1
+ cgso=4e-9 cgdo=4e-9 cgbo=16.667e-9 cbs=100.5e-7
+ cbd=100.5e-7)
.model nox nmos(vto=.75 kp=195u rd=.5 rs=.5 rg=1 rb=1
+ cgso=66.667e-12 cgdo=66.667e-12 cgbo=125e-9
+ cbs=2.34e-13 cbd=2.34e-13)
.model pix pmos(vto=-.75 kp=205.5u rd=1 rs=1 rg=1 rb=1
+ cgso=4e-12 cgdo=4e-12 cgbo=16.667e-9 cbs=2.534e-12
+ cbd=10.534e-12)
.model pox pmos(vto=-.75 kp=195u rd=.5 rs=.5 rg=1 rb=1
+ cgso=66.667e-12 cgdo=66.667e-12 cgbo=125e-9
+ cbs=2.538e-11 cbd=2.538e-11)
.ends NJU7044_s


```

MOSFET MODEL


Pspice model parameter	Model description
LEVEL	
L	Channel Length
W	Channel Width
KP	Transconductance
RS	Source Ohmic Resistance
RD	Ohmic Drain Resistance
VTO	Zero-bias Threshold Voltage
RDS	Drain-Source Shunt Resistance
TOX	Gate Oxide Thickness
CGSO	Zero-bias Gate-Source Capacitance
CGDO	Zero-bias Gate-Drain Capacitance
CBD	Zero-bias Bulk-Drain Junction Capacitance
MJ	Bulk Junction Grading Coefficient
PB	Bulk Junction Potential
FC	Bulk Junction Forward-bias Capacitance Coefficient
RG	Gate Ohmic Resistance
IS	Bulk Junction Saturation Current
N	Bulk Junction Emission Coefficient
RB	Bulk Series Resistance
PHI	Surface Inversion Potential
GAMMA	Body-effect Parameter
DELTA	Width effect on Threshold Voltage
ETA	Static Feedback on Threshold Voltage
THETA	Modility Modulation
KAPPA	Saturation Field Factor
VMAX	Maximum Drift Velocity of Carriers
XJ	Metallurgical Junction Depth
UO	Surface Mobility

Output Voltage Swing (V_{OH1})

Simulation result



Evaluation Circuit




Comparison Table


$R_L=10\text{ k}\Omega$ to 2.5 V	Measurement	Simulation	%Error
$V_{OH1}(\text{min})$ (V)	4.95	4.9957	-

Output Voltage Swing (V_{OL1})

Simulation result



Evaluation Circuit




Comparison Table


$R_L=10\text{ k}\Omega$ to 2.5 V	Measurement	Simulation	%Error
$V_{OL1}\text{ (MAX) (mV)}$	50	4.3511	-

Input Current

Simulation result



Evaluation Circuit




Comparison Table


	Measurement	Simulation	% Error
I_b (pA)	1	1	0
I_{IO} (pA)	1	1	0

Input Offset Voltage

Simulation result



Evaluation Circuit




Comparison Table


	Measurement	Simulation	%Error
V _{os} (mV)	10	9.812	-1.88

Open loop Voltage Gain

Simulation result



Evaluation Circuit




Comparison Table


	Measurement	Simulation	%Error
AV(dB)	90	89.996	-0.004
F0-db (MHz)	0.8	0.799783	-0.027

Common-Mode Rejection Ratio

Simulation result



Evaluation Circuit



$$\text{CMRR} = \text{AV}/\text{ACM}$$


$$= 20 \cdot \text{LOG}(31608.217/(3.1961\text{m}/100\text{u}))$$

Comparison Table


	Measurement	Simulation	%Error
CMRR (dB)	60	59.903	-0.162

Slew Rate

Simulation result



Evaluation Circuit



Comparison Table

	Measurement	Simulation	%Error
SR (V/us)	0.8	0.792	-1