


Device Modeling Report

COMPONENTS : OPERATIONAL AMPLIFIER (CMOS)
PART NUMBER : NJU7016
MANUFACTURER : NEW JAPAN RADIO



Bee Technologies Inc.

Spice Model



```

*$
*PART NUMBER: NJU7016
*MANUFACTURER: NEW JAPAN RADIO
*CMOS OPAMP
*All Rights Reserved Copyright (c) Bee Technologies Inc. 2008
.SUBCKT nju7016  OUT1 IN-1 IN+1 VSS IN+2 IN-2 OUT2 VDD
X_U1 IN+1 VSS IN-1 OUT1 VDD nju7016_s
X_U2 IN+2 VSS IN-2 OUT2 VDD nju7016_s
.ENDS nju7016
.SUBCKT nju7016_s  IN+ VSS IN- OUT VDD
M1      2 IN- 3 VDD MbreakPD3
M2      2 IN+ 4 VDD MbreakPD2
M3      VDD 1 2 VDD MbreakPD
M4      VDD 1 5 VDD MbreakPD
M5      VDD 1 6 VDD MbreakPD
M6      VDD 1 1 VDD MbreakPD
M7      5 5 VSS VSS MbreakND  W=3.2m  L=6u
M8      5 4 VSS VSS MbreakND3
M9      3 3 IN1 VSS MbreakND1
M10     4 3 IN2 VSS MbreakND1
M11     1 6 11 11 MbreakND  W=3.2m  L=6u
M12     6 6 VSS VSS MbreakND3
M13     7 5 VSS VSS MbreakND1
M14     VDD 7 7 VDD MbreakPD
M15     VDD 7 OUT VDD MbreakPD1
M16     OUT 4 VSS VSS MbreakND2
C1      1 IN+ 11p
C2      OUT 3 2.25p
R1      11 VSS 1.522k
R2      IN1 VSS 2.0k
R3      IN2 VSS 2.423k
I1      0 IN- 0.505p
I2      0 IN+ 1.5p
X_U1    VSS 3 DbreakZ
X_U2    VSS 4 DbreakZ

```

```

.model MbreakND NMOS (LEVEL=3 VTO=0.9 RS=10.000E-3 RD=10.000E-3
+ RDS=1.0000E6 TOX=2.0000E-6 CGSO=4.000E-12 CGDO=1.000E-12
+ CBD=1.000E-12 RG=5 RB=1.0000E-3 KP=10E-6)
.model MbreakND1 NMOS (LEVEL=3 L=6u W=0.165 VTO=1 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 CGSO=1.00E-12
+ CGDO=1.000E-12 CBD=10.000E-12 RG=5 RB=1.0000E-3 KP=10E-6)
.model MbreakND2 NMOS (LEVEL=3 L=6u W=2.3m VTO=0.9 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 CGSO=4.000E-12
+ CGDO=1.00E-12 CBD=1.000E-12 RG=5 RB=1.0000E-3 KP=10E-6)
.model MbreakND3 NMOS (LEVEL=3 L=6u W=3.2m VTO=0.9 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 CGSO=1.000E-12
+ CGDO=1.000E-12 CBD=1.000E-12 RG=5 RB=1.0000E-3 KP=10E-6)
.model MbreakPD PMOS (LEVEL=3 L=6u W=0.23 VTO=-1 RS=10.000E-3
+ RD=10.000E-3 RDS=1.0000E6 TOX=2.0000E-6 CGSO=4.000E-12
+ CGDO=1.000E-12 CBD=1.000E-12 RG=5 RB=1.0000E-3 KP=1E-6)
.MODEL MbreakPD1 PMOS (LEVEL=3 L=6u W=0.0639 VTO=-0.9
+ RS=10.000E-3 RD=10.000E-3 RDS=1.00E6 TOX=2.0000E-6
+ CGSO=1.1800E-11 CGDO=1.000E-12 CBD=1.000E-12 RG=5
+ RB=1.0000E-3 KP=1E-6)
.MODEL MbreakPD2 PMOS (LEVEL=3 L=6u W=0.00017 VTO=-1.4
+ RS=10.000E-3 RD=10.00E-3 RDS=1.069e6 TOX=2.0000E-6
+ CGSO=1.000E-9 CGDO=1.000E-12 CBD=1.00E-12
+ RG=5 RB=1.0000E-3 KP=1E-6)
.MODEL MbreakPD3 PMOS (LEVEL=3 L=6u W=0.000182 VTO=-1.4
+ RS=10.000E-3 RD=10.00E-3 RDS=1.00E6 TOX=2.0000E-6
+ CGSO=1.000E-12 CGDO=1.000E-12 CBD=1.00E-12 RG=5
+ RB=1.0000E-3 KP=1E-6)
.ENDS nju7016_s
.SUBCKT DbreakZ A K
D1 A K DF
DZ A2 A DR
VZ K A2 1
.MODEL DF D
.MODEL DR D
.ENDS DbreakZ
*$


```

MOSFET MODEL


| Pspice model parameter | Model description |
|------------------------|--|
| LEVEL | |
| L | Channel Length |
| W | Channel Width |
| KP | Transconductance |
| RS | Source Ohmic Resistance |
| RD | Ohmic Drain Resistance |
| VTO | Zero-bias Threshold Voltage |
| RDS | Drain-Source Shunt Resistance |
| TOX | Gate Oxide Thickness |
| CGSO | Zero-bias Gate-Source Capacitance |
| CGDO | Zero-bias Gate-Drain Capacitance |
| CBD | Zero-bias Bulk-Drain Junction Capacitance |
| MJ | Bulk Junction Grading Coefficient |
| PB | Bulk Junction Potential |
| FC | Bulk Junction Forward-bias Capacitance Coefficient |
| RG | Gate Ohmic Resistance |
| IS | Bulk Junction Saturation Current |
| N | Bulk Junction Emission Coefficient |
| RB | Bulk Series Resistance |
| PHI | Surface Inversion Potential |
| GAMMA | Body-effect Parameter |
| DELTA | Width effect on Threshold Voltage |
| ETA | Static Feedback on Threshold Voltage |
| THETA | Modility Modulation |
| KAPPA | Saturation Field Factor |
| VMAX | Maximum Drift Velocity of Carriers |
| XJ | Metallurgical Junction Depth |
| UO | Surface Mobility |

Output Voltage Swing

Simulation result



Evaluation Circuit




Comparison Table


| | Measurement | Simulation | %Error |
|--------------|-------------|------------|--------|
| V_{OM} (V) | 2.9 | 2.901 | 0.034 |

Input Current

Simulation result



Evaluation Circuit




Comparison Table


| | Measurement | Simulation | % Error |
|---------------|-------------|------------|---------|
| I_b (pA) | 1 | 1.002 | 0.2 |
| I_{os} (pA) | 1 | 0.995 | -0.5 |

Input Offset Voltage

Simulation result



Evaluation Circuit




Comparison Table


| | Measurement | Simulation | %Error |
|---------------|-------------|------------|--------|
| V_{os} (mV) | 10 | 9.995 | -0.05 |

Open loop Voltage Gain

Simulation result



Evaluation Circuit




Comparison Table


| | Measurement | Simulation | %Error |
|---------|-------------|------------|--------|
| Av (dB) | 70 | 69.7 | -0.429 |

Unity Gain Frequency

Simulation result



Evaluation Circuit




Comparison Table


| $A_v=40\text{dB}, C_L=10\text{pF}$ | Measurement | Simulation | %Error |
|------------------------------------|-------------|------------|--------|
| Ft(MHz) | 1 | 1 | 0 |

Common-Mode Rejection Ratio

Simulation result



Evaluation Circuit




$$\text{CMRR} = \text{AV}/\text{ACM}$$

Comparison Table


| | Measurement | Simulation | %Error |
|------------------|-------------|------------|--------|
| CMRR (dB) | 65 | 62.537 | -3.789 |

Slew Rate

Simulation result



Evaluation Circuit



Comparison Table

| | Measurement | Simulation | % Error |
|-----------|-------------|------------|---------|
| SR (V/us) | 2.4 | 2.350 | -2.083 |